
Treść wykładu

Treść wykładu

Treść wykładu

Treść wykładu

Przykłady
● Odlot samolotu (zegar i czas zdarzeń)

● Funkcja z pętlą for (analiza kodu programu)

Treść wykładu

Praca interaktywna
● Czynności wstępne

● Weryfikacja modelu

● Symulacja modelu

Modelowanie systemu

● Modelowanie pośrednie

● Modelowanie bezpośrednie

● Założenia FAIRNESS

● Niedeterminizm



8

Modelowanie systemu

Modelowanie 
pośrednie
MODULE main

VAR a : boolean;
b : 0..4;

ASSIGN init(a):= TRUE;
next(a):= !a;
init(b):= {0,2,4};
next(b):= case

next(a) : {0,2,4};
!next(a) : {1,3}; 

esac;

CTLSPEC AG(a -> b in {0,2,4})

INVARSPEC (!a -> b in {1,3})

Modelowanie 
bezpośrednie
MODULE main

VAR a : boolean;
b : 0..4;

INIT a = TRUE &
b in {0,2,4};

TRANS next(a) = !a;
TRANS next(b) in case

next(a) : {0,2,4};
!next(a) : {1,3}; 

esac;

CTLSPEC AG(a -> b in {0,2,4})

INVARSPEC (!a -> b in {1,3})

Modelowanie systemu

Modelowanie pośrednie
● Działanie automatu jest zdefiniowane przez podanie 

początkowych i następnych wartości zmiennych stanowych.

● Przykład:

ASSIGN init(a):= TRUE;
next(a):= !a;
init(b):= {0,2,4};
next(b):= case

next(a) : {0,2,4};
!next(a) : {1,3};

esac;

● Operator init definiuje początkową wartość zmiennej.

● Operator next definiuje wartość zmiennej w następnym stanie.

Modelowanie systemu

Modelowanie pośrednie
● Jeśli nie podano początkowej wartości dla zmiennej, otrzyma 

ona dowolną wartość z jej zakresu wartości.
(Istnieje min. 1 stan początkowy.)

● Jeśli nie podano następnej wartości dla zmiennej, otrzyma ona 
dowolną wartość z jej zakresu wartości.

(Dla każdego stanu istnieje min. 1 stan następny.)

Uwaga
● Każdy model zdefiniowany pośrednio można zdefiniować 

bezpośrednio.

● Nie każdy model zdefiniowany bezpośrednio można
zdefiniować pośrednio. 

Modelowanie systemu

Modelowanie bezpośrednie
● Działanie automatu jest zdefiniowane przez podanie wyrażeń 

logicznych.

● Wyrażenia logiczne określają:
● stany początkowe,
● stany osiągalne,

● przejścia między stanami.

● Skutki braku wyrażeń lub ich wzajemnej sprzeczności:
● pusty zbiór stanów początkowych,

● nieosiągalne stany,
● brak stanów osiągalnych.

Modelowanie bezpośrednie
● Określenie początkowej wartości zmiennych:

INIT wyrażenie_logiczne

● Wyrażenie podane po INIT określa początkowe wartości zmiennych.

● Przykład określenia wartości zmiennych a i b:

INIT a = TRUE &
b in {0,2,4}

● Jeśli nie podano początkowej wartości dla zmiennej,
otrzyma ona dowolną wartość z jej zakresu wartości.

● Jeśli podano nieprawdziwe wyrażenie, to brak stanów 
początkowych (weryfikacja modelu może być błędna).

● Nie można używać operatora next.

Modelowanie systemu

Modelowanie systemu

Modelowanie bezpośrednie
● Określenie osiągalnych stanów przez niezmienniki stanów:

INVAR wyrażenie_logiczne

● Wyrażenie podane po INVAR określa wartości zmiennych,
które cechują każdy stan.

● Przykład określenia wartości zmiennych a i b:

INVAR a=TRUE | a=FALSE

INVAR !a -> b in {1,3}

● Jeśli podano nieprawdziwe wyrażenie, to brak stanów 
osiągalnych (weryfikacja modelu może być błędna).

● Definiowanie niezmienników nie jest obowiązkowe.

● Nie można używać operatora next.

Modelowanie systemu

Modelowanie bezpośrednie
● Określenie dozwolonych przejść między stanami:

TRANS wyrażenie_logiczne

● Wyrażenie podane po TRANS określa dozwolone wartości zmiennych 
w następnym stanie.

● Przykład określenia następnych wartości zmiennych a i b:

TRANS next(a) = !a;
TRANS next(b) in case

next(a) : {0,2,4};
!next(a) : {1,3};

esac;

● Jeśli podano nieprawdziwe wyrażenie, to może nie być 
następnego stanu (weryfikacja modelu może być błędna).

Modelowanie systemu

Modelowanie bezpośrednie
● INVAR czy INIT połączony z TRANS?

● 1. sposób – niezmiennie a = 1:

INVAR a=1

● 2. sposób – w początkowym i każdym następnym stanie a = 1:

INIT a=1
TRANS next(a)=1

● Efekt wydaje się ten sam, ale 1. sposób jest efektywniejszy.

– W takiej sytuacji zaleca się podanie niezmiennika.

Modelowanie systemu

Założenia FAIRNESS
● Założenie JUSTICE wyrażenie

● Alternatywnie: FAIRNESS wyrażenie

● Weryfikacja modelu obejmie tylko te ścieżki, na których wyrażenie
jest prawdziwe nieskończenie wiele razy, np.:

VAR a : boolean;
JUSTICE !a

● Odpowiada to formule AG(AF(Ø(a))).

● W wyrażeniu nie można używać operatora next.



17

Modelowanie systemu

Założenia FAIRNESS
● Założenie COMPASSION (wyrażenie1,wyrażenie2)

● Weryfikacja modelu obejmie tylko te ścieżki, na których:

– jeśli wyrażenie1 jest prawdziwe nieskończenie wiele razy,
– to wyrażenie2 też jest prawdziwe nieskończenie wiele razy

na tych samych ścieżkach, np.:

VAR a : boolean;
b : boolean;

COMPASSION (!a,!b)

● Odpowiada to formule AG(AG(AF(Ø(a))) Þ AG(AF(Ø(b)))).

● W wyrażeniach nie można używać operatora next.

● NuSMV jeszcze nie w pełni wspiera COMPASSION.

Modelowanie systemu

Niedeterminizm
● Definicja zmiennej wymaga podania zbioru jej wartości, np.:

VAR
a : 0..10;
b : {s1, s2, s3};

● Jeśli żadna instrukcja nie przypisuje wartości do zmiennej,
to zmienna otrzymuje dowolną wartość ze zbioru jej wartości.

● Jeśli instrukcja przypisuje do zmiennej podzbiór zbioru jej 
wartości, to zmienna otrzymuje dowolną wartość z tego 
podzbioru, np:

b := {s1,s3}

Błędy w modelowaniu 
systemu

● Różne definicje zmiennej

● Rekursywna definicja zmiennej

● Wzajemna zależność definicji zmiennych

● Sprzeczności w wyrażeniach INIT, INVAR i TRANS

Błędy w modelowaniu systemu

Różne definicje zmiennej
● Zmienna powinna mieć tylko jedną definicję określającą jej 

wartość dla danego stanu:
● źle: init(a) := TRUE;

init(a) := FALSE;

● źle: b := a;
b := a+1;

● źle: init(c) := a;
c := b;

● dobrze: init(a) := {TRUE,FALSE};

Błędy w modelowaniu systemu

Rekursywna definicja zmiennej
● Wartość zmiennej nie może zależeć od jej wartości z tego 

samego stanu:
● źle: a := a+1;

● źle: next(a) := next(a)+1;

● Ale może zależeć od jej wartości z następnego stanu:
● dobrze: next(a) := a+1;

Błędy w modelowaniu systemu

Wzajemna zależność definicji zmiennych
● Wartości zmiennych w tym samym stanie nie mogą być 

wzajemnie zależne:
● źle: a := b+1;

b := a-1;

● źle: next(a) := next(b);
next(b) := next(a);

● Ale wartości zmiennych z różnych stanów mogą być wzajemnie 
zależne:

● dobrze: next(a) := b;
next(b) := a;

● dobrze: next(a) := next(b);
next(b) := a;

Błędy w modelowaniu systemu

Sprzeczności w wyrażeniach INIT, INVAR i TRANS
● Jeśli podano nieprawdziwe wyrażenie INIT ,

to brak stanów początkowych.

● Jeśli podano nieprawdziwe wyrażenie INVAR ,
to brak stanów osiągalnych.

● Jeśli podano nieprawdziwe wyrażenie TRANS ,
to może nie być następnego stanu.

● Te błędy są raportowane przez NuSMV.

● Te błędy mogą doprowadzić do błędnej weryfikacji modelu.

Weryfikowanie systemu

● Możliwości

● Rodzaje własności do zweryfikowania

● Obliczenie minimalnej i maksymalnej ścieżki stanów

Weryfikowanie systemu

Możliwości
● Weryfikacja jest automatyczna.

● Specyfikację systemu opisuje się formułami logiki temporalnej.

● Dostępne logiki: LTL, LTL–, CTL, RTCTL (z dolnym i górnym 
ograniczeniem operatorów temporalnych) i PSL.

● Dozwolone są wszystkie poprawnie zbudowane formuły.

● Każda formuła weryfikowana jest niezależnie od pozostałych.

● Weryfikacja formuły zwraca true lub false.

● Wynikowi false towarzyszy kontrprzykład (ścieżka stanów),
jeśli można go wygenerować.

● Można wyliczyć długość minimalnej i maksymalnej ścieżki
między dwoma określonymi stanami.

Weryfikowanie systemu

Rodzaje własności do zweryfikowania
● Własności opisane w logice LTL (dotyczące czasu liniowego):

LTLSPEC formuła_LTL
● Własności opisane w logice CTL (dotyczące czasu rozgałęzionego):

CTLSPEC formuła_CTL
● Własności opisane w logice LTL–, PSL, RTCTL.

● Niezmienniki (dotyczące każdego stanu modelu):

INVARSPEC wyrażenie_logiczne

Weryfikowanie systemu

Obliczenie minimalnej i maksymalnej ścieżki stanów
● Polecenie COMPUTE oblicza długość ścieżki (liczbę stanów) 

między dwoma określonymi stanami.

● Opis stanu to wyrażenie logiczne określające wartości 
wybranych zmiennych stanowych w tym stanie.

● Obliczenie minimalnej ścieżki:

COMPUTE MIN[stan1,stan2]

● Obliczenie maksymalnej ścieżki:

COMPUTE MAX[stan1,stan2]

● Wynikiem jest liczba stanów lub INFINITY 
(nieskończoność).

Przykłady

● Odlot samolotu (zegar i czas zdarzeń)

● Funkcja z pętlą for (analiza kodu programu)

Przykłady

Odlot samolotu (zegar i czas zdarzeń)
● Opis sytuacji:

● Pas startowy krzyżuje się z drogą kołowania.
● Samolot rozpoczyna ruch przed skrzyżowaniem, przyspieszając.
● Samolot, przyspieszając, osiąga prędkość V1 (po 6..8 sekundach),

po czym odlatuje (po 1..3 sekundach).
● Odlot samolotu może nastąpić przed, na lub po skrzyżowaniu.
● Intruz może pojawić się na skrzyżowaniu w dowolnym momencie.
● Intruz, gdy pojawi się na skrzyżowaniu, nie znika z niego.
● Jeśli samolot przyspiesza przed lub na skrzyżowaniu, gdzie pojawia się intruz,

to zwalnia, jeśli jego prędkość < V1.
● Zwalniający samolot staje (po 3..4 sekundach) przed, na lub po skrzyżowaniu.
● Jeśli samolot i intruz są jednocześnie na skrzyżowaniu, to może nastąpić kolizja.
● Stany końcowe: samolot odlatuje, samolot stoi, jest kolizja.

Przykłady

Odlot samolotu (zegar i czas zdarzeń)
MODULE main

VAR
--położenie samolotu względem skrzyżowania
gdzie : {przed, na, po};
--rodzaj ruchu samolotu
ruch: {przyspiesza, zwalnia, stoi, odlatuje};
--zegar dla czasu ruchu samolotu
--(zerowany w momencie zmiany rodzaju ruchu)
t : 0..9;
--intruz na skrzyżowaniu
intruz : boolean;
--kolizja z intruzem
kolizja : boolean;
--prędkość samolotu >= V1 (zakaz hamowania)
v1 : boolean;

Przykłady

Odlot samolotu (zegar i czas zdarzeń)
--STAN POCZĄTKOWY

INIT
--samolot jest przed skrzyżowaniem
gdzie = przed &
--samolot przyspiesza
ruch = przyspiesza &
--rozpoczyna się czas przyspieszania
t = 0 &
--nie ma intruza na skrzyżowaniu
intruz = FALSE &
--nie ma kolizji
kolizja = FALSE &
--prędkość samolotu < V1
v1 = FALSE

Przykłady

Odlot samolotu (zegar i czas zdarzeń)
--ZACHOWANIE ZEGARA t

TRANS next(t) in case
--zerowanie zegara przy rozpoczęciu odlotu
ruch = przyspiesza & next(ruch) = odlatuje : 0;
--zerowanie zegara przy rozpoczęciu zwalniania
ruch = przyspiesza & next(ruch) = zwalnia : 0;
--zerowanie zegara przy rozpoczęciu stania
ruch = zwalnia & next(ruch) = stoi : 0;
--w innym przypadku z każdą zmianą stanu automatu
--upływa jedna jednostka czasu
TRUE : (t + 1) mod 10; esac;

Przykłady

Odlot samolotu (zegar i czas zdarzeń)
--ZACHOWANIE INTRUZA

TRANS next(intruz) in case
--intruz może pojawić się w dowolnym momencie
!intruz : {FALSE,TRUE};
--intruz nie może zniknąć ze skrzyżowania,
--jeśli już na nim jest
TRUE : intruz; esac;

Przykłady

Odlot samolotu (zegar i czas zdarzeń)
--ZACHOWANIE PRĘDKOŚCI V1

TRANS next(v1) in case
--nie można osiągnąć V1 w czasie t < 6
!v1 & ruch = przyspiesza & next(t)<6 : FALSE;
--można osiągnąć V1 w czasie t < 8
!v1 & ruch = przyspiesza & next(t)<8 : {TRUE,FALSE};
--V1 osiągnięta jest najpóźniej w czasie t = 8
!v1 & ruch = przyspiesza & next(t)=8 : TRUE;
--osiągnięta prędkość V1 nie maleje
TRUE : v1; esac;

Przykłady

Odlot samolotu (zegar i czas zdarzeń)
--ZACHOWANIE KOLIZJI

--kolizja niemożliwa, jeśli nie ma intruza
--lub samolot jest przed skrzyżowaniem
INVAR !intruz | gdzie = przed -> !kolizja;

TRANS next(kolizja) in case
--jeśli jest kolizja, to nie zniknie
kolizja : TRUE;
--jeśli nie ma kolizji, to jest możliwa,
--jeśli intruz i samolot są na skrzyżowaniu
intruz & gdzie = na : {FALSE, TRUE};
--pozostałe stany nie wpływają na kolizję
TRUE : kolizja; esac;

Przykłady

Odlot samolotu (zegar i czas zdarzeń)
--ZACHOWANIE POŁOŻENIA SAMOLOTU

TRANS next(gdzie) in case
--stojący lub odlatujący samolot nie zmienia swego położenia
--(stan końcowy)
ruch = stoi | ruch = odlatuje : gdzie;
--samolot będący przed skrzyżowaniem może wjechać na nie
gdzie = przed : {przed, na};
--samolot będący na skrzyżowaniu może zjechać z niego
gdzie = na : {na, po};
--samolot będący po skrzyżowaniu nie zmienia swego położenia
gdzie = po: po; esac;

Przykłady

Odlot samolotu (zegar i czas zdarzeń)
--ZACHOWANIE RUCHU SAMOLOTU (1)

TRANS next(ruch) in case
--samolot przyspieszający z prędkością >= v1
--nie może odlecieć jeśli jest kolizja
--(brak zmiany ruchu, stan końcowy )
ruch = przyspiesza & v1 & kolizja : przyspiesza;
--samolot przyspieszający z prędkością >= v1
--nie może odlecieć w czasie t < 1
ruch = przyspiesza & v1 & next(t)<1 : przyspiesza;
--samolot przyspieszający z prędkością >= v1
--może odlecieć w czasie t < 3 (jeśli nie ma kolizji)
ruch = przyspiesza & v1 & next(t)<3 :{przyspiesza, odlatuje};
--samolot przyspieszający z prędkością >= v1 odlatuje
--najpóźniej w czasie = 3 (jeśli nie ma kolizji)
ruch = przyspiesza & v1 & next(t=3) : odlatuje;

Przykłady

Odlot samolotu (zegar i czas zdarzeń)
--ZACHOWANIE RUCHU SAMOLOTU (2)

-- ...
--samolot przyspieszający z prędkością < v1
--nadal przyspiesza (jeśli nie ma intruza)
ruch =przyspiesza & !v1 & !intruz: przyspiesza;
--samolot przyspieszający z prędkością < v1
--zwalnia, jeśli na skrzyżowaniu jest intruz
ruch = przyspiesza & !v1 & intruz : zwalnia;

Przykłady

Odlot samolotu (zegar i czas zdarzeń)
--ZACHOWANIE RUCHU SAMOLOTU (3)

-- ...
--samolot zwalniający nie stanie w czasie t<3
ruch = zwalnia & next(t) < 3 : zwalnia;
--samolot zwalniający może stanąć w czasie t<4
ruch = zwalnia & next(t) < 4 : {zwalnia, stoi};
--samolot zwalniający stanie najpóźniej w czasie t=4
ruch = zwalnia & next(t) = 4 : stoi;
--samolot stojący lub odlatujący nie zmienia swego ruchu
ruch = stoi | ruch = odlatuje : ruch;
--pozostałe stany nie wpływają na ruch
TRUE : ruch; esac;

Przykłady

Odlot samolotu (zegar i czas zdarzeń)
● Weryfikacja zachowania zegara:

-- Inkrementacja zegara z każdą zmianą stanu(mod 10)
CTLSPEC AG(t=0 -> AX(t=1))
CTLSPEC AG(t=9 -> AX(t=0))
COMPUTE MIN[t=0,t=1] --ma być 1
COMPUTE MAX[t=0,t=1] --ma być 1

-- Zmiana ruchu samolotu powoduje wyzerowanie zegara
--(np. zmiana zwalniania na stanie)

CTLSPEC AG(ruch=zwalnia & AX(ruch=stoi) -> AX(t=0))
CTLSPEC AG(ruch=zwalnia & AX(ruch=stoi) & t!=0 -> AX(t=0))

Przykłady

Odlot samolotu (zegar i czas zdarzeń)
● Weryfikacja zachowania prędkości V1:

--Przyspieszający samolot osiąga prędkość V1
--po 6..8 sekundach

CTLSPEC EF(!v1 & ruch=przyspiesza -> EX v1)
CTLSPEC AG(!v1 & ruch=przyspiesza & AX t=8 -> AX v1)
CTLSPEC AG(!v1 & ruch=przyspiesza & AX t<6 -> AX !v1)
CTLSPEC AG(!v1 & ruch=przyspiesza & AX t>=6 & AX t<8

-> EX !v1) --prawidłowo
CTLSPEC AG(!v1 & ruch=przyspiesza & AX t>=6 & AX t<8

-> AX !v1) --nieprawidłowo

Przykłady

Funkcja z pętlą for (analiza kodu programu)
● Ta funkcja oblicza sumę liczb od 0 do x:

int suma(int x){
 int s = 0; //suma liczb
 for(int i = 0; i <= x; i++)
  s = s + i;
 return s;}

● Cel – przeanalizuj pętlę for, np.:

● znajdź maksymalną liczbę wykonań pętli,

● sprawdź osiągalność s=21 dla x=6.

Przykłady

Funkcja z pętlą for (analiza kodu programu)
● Etykiety wykonywanych operacji:

int suma(int x){
 int s = 0; //suma liczb e0 (tę można pominąć)
 for(int i = 0; i <= x; i++) e1
  s = s + i; e2
 return s;} e3

● Model operacji:

MODULE main
FROZENVAR x : 0..6;   --stała x o losowej wartości 0 do 6
VAR e : {e1, e2, e3}; --etykiety operacji
    s : 0..21;        --suma liczb
    i : 0..7;         --iterator pętli
    c : 0..6;         --licznik wykonań pętli
    a : 1..3;         --flaga osiągnięcia deadlocka

Przykłady

Funkcja z pętlą for (analiza kodu programu)
int suma(int x){
 int s = 0; //suma liczb
 for(int i = 0; i <= x; i++) e1
  s = s + i; e2
 return s; e3

● Model operacji:

ASSIGN
--sterowanie operacjami:
init(e) := e1;  --rozpoczęcie w e1
next(e) := case
 e = e1 & i <= x : e2; --przejście do e2 z e1 jeśli i<=x

--(aby wykonać pierwszą operację z bloku for)
 e = e1 & i > x : e3; --przejście do e3 z e1 jeśli i>x

--(aby nie wykonywać operacji z bloku for)
 e = e2 : e1; --powrót do e1 z e2

--(wykonano ostatnią operację z bloku for)
 e = e3 : e3; esac; --brak wyjścia w e3 (koniec programu)

Przykłady

Funkcja z pętlą for (analiza kodu programu)
int suma(int x){
 int s = 0; //suma liczb
 for(int i = 0; i <= x; i++) e1
  s = s + i; e2
 return s; e3

● Model operacji:

ASSIGN
--zmienność s:
init(s) := 0; --początkowo s=0
next(s) := case
 s + i > 21 : s; --brak zmiany s

--(aby nie przekroczyć jej zakresu)
 next(e) = e2 : s + i; --dodanie i do s (w operacji e2)
 TRUE : s; esac; --brak zmiany s

--(w pozostałych przypadkach)

Przykłady

Funkcja z pętlą for (analiza kodu programu)
int suma(int x){
 int s = 0; //suma liczb
 for(int i = 0; i <= x; i++) e1
  s = s + i; e2
 return s; e3

● Model operacji:

ASSIGN
--zmienność i:
init(i) := 0; --początkowo i=0
next(i) := case
 i + 1 > 7 : i; --brak zmiany i

--(aby nie przekroczyć jej zakresu)
 next(e) = e1 : i + 1; --inkrementacja i

--(w operacji e1, czyli po e2)
 TRUE : i; esac; --brak zmiany i

--(w pozostałych przypadkach)

Przykłady

Funkcja z pętlą for (analiza kodu programu)
int suma(int x){
 int s = 0; //suma liczb
 for(int i = 0; i <= x; i++) e1
  s = s + i; e2
 return s; e3

● Model operacji:

ASSIGN
--zmienność c:
init(c) := 0; --początkowo c=0
next(c) := case
 c + 1 > 6 : c; --brak zmiany c

--(aby nie przekroczyć jej zakresu)
 next(e) = e2 : c + 1; --inkrementacja c

--(w operacji e2, czyli bloku for)
 TRUE : c; esac; --brak zmiany c

--(w pozostałych przypadkach)

Przykłady

Funkcja z pętlą for (analiza kodu programu)
int suma(int x){
 int s = 0; //suma liczb
 for(int i = 0; i <= x; i++) e1
  s = s + i; e2
 return s; e3

● Model operacji:

ASSIGN
--zmienność a:
init(a) := 1; --początkowo a=1 (będzie następny stan)
next(a) := case
 a = 1 & e = next(e) & s = next(s) & i = next(i) & c = next(c) : 2;

--ustawienie a na 2
--(jeśli pierwszy raz nie nastąpi zmiana stanu)

 a = 2 : 3; --ustawienie a na 3
--(aby 2 było tylko w jednym stanie)

 TRUE : a; esac; --brak zmiany a
--(w pozostałych przypadkach)

Przykłady

Funkcja z pętlą for (analiza kodu programu)
int suma(int x){
 int s = 0; //suma liczb
 for(int i = 0; i <= x; i++) e1
  s = s + i; e2
 return s; e3

● Weryfikacja modelu:

● Czy flaga a działa prawidłowo?

CTLSPEC AF(a=2 & AX(AG(a=3))) true

Przykłady

Funkcja z pętlą for (analiza kodu programu)
int suma(int x){
 int s = 0; //suma liczb
 for(int i = 0; i <= x; i++) e1
  s = s + i; e2
 return s; e3

● Analiza kodu:

● Jaka jest maksymalna długość ścieżki stanów
do stanu końcowego?

COMPUTE MAX[a=1, a=2] 16

● Jaka jest maksymalna liczba wykonań pętli for?

– wygeneruj ścieżkę maksymalnej długości,
– sprawdź końcową wartość licznika c na tej ścieżce.
CTLSPEC !EBF 16..16 (a=2) false c = 6

Przykłady

Funkcja z pętlą for (analiza kodu programu)
int suma(int x){
 int s = 0; //suma liczb
 for(int i = 0; i <= x; i++) e1
  s = s + i; e2
 return s; e3

● Analiza kodu:

● Czy pętla for może się wykonać się 6 razy?

CTLSPEC EF(c=6) false
CTLSPEC !EF(c=6) false Jak to jest możliwe?

● Czy suma s = 21 jest osiągalna?

CTLSPEC EF(s=21) false
CTLSPEC !EF(s=21) false Jak to jest możliwe?

Przykłady

Funkcja z pętlą for (analiza kodu programu)
int suma(int x){
 int s = 0; //suma liczb
 for(int i = 0; i <= x; i++) e1
  s = s + i; e2
 return s; e3

● Analiza kodu:

● Jeśli istnieje więcej niż 1 początkowy stan, to sprawdzanie osiągalności 
danego stanu formułą EF(sprawdzany_stan) zwróci false, jeśli ten 
stan nie jest osiągalny z każdego stanu początkowego, nawet jeśli jest 
osiągalny z pewnego stanu początkowego.

– rozwiązanie 1: wybierz początkowy stan:
początkowy_stan -> EF(sprawdzany_stan)

– rozwiązanie 2: sprawdź nieosiągalność stanu:
!EF(sprawdzany_stan)

Przykłady

Funkcja z pętlą for (analiza kodu programu)
int suma(int x){
 int s = 0; //suma liczb
 for(int i = 0; i <= x; i++) e1
  s = s + i; e2
 return s; e3

● Analiza kodu:

● Czy suma s = 21 jest osiągalna?

CTLSPEC x=6 -> EF(s=21) true
CTLSPEC !EF(s=21) false

● Czy suma s = 21 zostanie osiągnięta?

CTLSPEC x=6 -> AF(s=21) true

Praca interaktywna

● Czynności wstępne

● Weryfikacja modelu

● Symulacja modelu

● Ponowne rozpoczęcie i zakończenie pracy

● Wykonanie skryptu z poleceniami

● Opis czynności wykonywanych przez NuSMV

Praca interaktywna

Czynności wstępne
● Rozpocznij pracę z plikiem .smv w trybie interaktywnym:

NuSMV -int plik

● Wczytaj model systemu:

read_model

● Utwórz moduły i procesy:

flatten_hierarchy

● Pokaż listę zmiennych wejściowych i stanowych:
(opcjonalnie)

show_vars

Praca interaktywna

Czynności wstępne
● Pokaż zmienne zależne od podanego wyrażenia:

(opcjonalnie)

show_dependencies -e wyrażenie

● Utwórz zmienne do kompilacji modelu do postaci BDD 
(binarnych diagramów decyzyjnych):

encode_variables

● Zapisz kolejność zmiennych do pliku:
(opcjonalnie)

write_order

● Skompiluj model do postaci BDD:

build_model

Praca interaktywna

Czynności wstępne
● Zainicjuj systemu gotowy do weryfikacji:

go

● Wczytaj i skompiluj model do BDD, zweryfikuj model i oblicz 
zbiór osiągalnych stanów:

process_model

● Oblicz zbiór osiągalnych stanów:

compute_reachable

● Pokaż osiągalne stany:
(opcjonalnie)

print_reachable_states -v

Praca interaktywna

Weryfikacja modelu
● Pokaż wszystkie własności:

show_property

● Dodaj własność danego typu do weryfikacji:

add_property -typ -p ”formuła”

● Dodaj własność do weryfikacji w kontekście danego modułu:

add_property -typ -p ”formuła IN moduł”

Typ: c (formuła CTL), l (formuła LTL), s (formuła PSL), i (niezmiennik),
q (wyliczenie ścieżki).

Praca interaktywna

Weryfikacja modelu
● Zweryfikuj formulę CTL o podanym numerze:

check_ctlspec -n numer

● Zweryfikuj podaną formułę CTL:

check_ctlspec -p ”formuła”

● Zweryfikuj podaną formułę CTL w kontekście danego modułu:

check_ctlspec -p ”formuła IN moduł”

Analogicznie dla specyfikacji LTL: check_ltlspec

Praca interaktywna

Weryfikacja modelu
● Sprawdź możliwość blokady systemu (deadlock):

check_fsm

● Wylicz długość ścieżki między określonymi stanami
(dla wyrażenia o podanym numerze):

check_compute -n numer

● Wylicz długość minimalnej ścieżki między określonymi stanami:

check_compute -p ”MIN[stan1,stan2]”

● Wylicz długość maksymalnej ścieżki między określonymi 
stanami w kontekście danego modułu:

check_compute -p ”MAX[stan1,stan2] IN moduł”

Praca interaktywna

Weryfikacja modelu
● Zweryfikuj niezmiennik o podanym numerze:

check_invar -n numer

● Zweryfikuj podany niezmiennik:

check_invar -p ”niezmiennik”

● Zweryfikuj podany niezmiennik w kontekście danego modułu:

check_invar -p ”niezmiennik IN moduł”

Praca interaktywna

Symulacja modelu
● Wybierz losowo początkowy stan:

pick_state -r

● Wybierz początkowy stan z listy dostępnych stanów:

pick_state -i

● Wybierz początkowy stan z ograniczonej listy dostępnych 
stanów – ze stanów spełniających podane wyrażenie:

pick_state -i -c ”wyrażenie”



63

Praca interaktywna

Symulacja modelu
● Przeprowadź symulację od wybranego stanu:

simulate [-p|-v] [-r|-i] [-k liczba_stanów]

● wyświetlaj tylko zmienione zmienne stanu: -p

● wyświetlaj wszystkie zmienne stanu: -v

● losowo wybieraj dostępne stany: -r

● ręcznie wybieraj dostępne stany: -i

● podawaj długość ścieżki stanów (np. 4): -k 4

(Domyślnie symulacja obejmuje 10–stanowe ścieżki.)

● przykłady:

simulate -p -r -k 5
simulate -v -i

Praca interaktywna

Symulacja modelu
Analiza wybranej ścieżki stanów:

● Ścieżki stanów tworzone są w wyniku negatywnej weryfikacji 
formuły i w wyniku symulacji.

● Pokaż wygenerowane ścieżki stanów:
● wszystkie: show_traces -v -a

● wybraną: show_traces -v nr_ścieżki

● wybraną ze stanami (od – do):
show_traces -v nr_ścieżki.nr_stanu_od:nr_stanu_do

● Pokaż liczbę wygenerowanych ścieżek:

show_traces -t

Praca interaktywna

Symulacja modelu
Analiza wybranej ścieżki stanów:

● Przejdź do wybranego stanu wybranej ścieżki:

goto_state nr_ścieżki.nr_stanu

● Pokaż opis bieżącego stanu bieżącej ścieżki:

print_current_state -v

Praca interaktywna

Ponowne rozpoczęcie i zakończenie pracy

● Ponownie rozpocznij pracę (reset ustawień):

reset

● Zakończ pracę (reset ustawień):

quit

Praca interaktywna

Wykonanie skryptu z poleceniami
● Automatycznie wykonaj określoną sekwencję poleceń z pliku:

NuSMV -source plik

● Wystąpienie błędu uniemożliwia wykonanie dalszych poleceń.

Opis czynności wykonywanych przez NuSMV
● Ustaw szczegółowość opisu wykonywanych czynności:

NuSMV -v N -int plik

(N – poziom opisu: od 0 (brak) do 4)



Koniec

Literatura:
● K.L. McMillan, „The SMV system”, 2001

● R. Cavada et al. „NuSMV 2.5 User Manual”, 2010
● R. Cavada et al. „NuSMV 2.5 Tutorial”

